



地基基础计算与程序设计

王 国 体 主 编



中 国 科 学 技 术 大 学 出 版 社

地基基础计算与程序设计

王国体 主编



中国科学技术大学出版社

内 容 简 介

本书介绍了计算机在地基基础学科各个领域,包括土工测试数据整理、基础底面积设计和基础构件的数值计算等方面的应用。全书分九章,包括室内土工测试、基础沉降、地基承载力、土坡(坝)稳定分析、挡土墙计算、基础群设计和有限元法、差分法计算文克尔地基梁与板等基础构件的计算理论、计算方法和应用程序。源程序采用 BASIC 语言编写,结构性强,通用性广,使用方便,可在各类计算机系统上运行。

本书可作为高等工科院校土木、建筑、交通、水利等专业师生的教学参考书或选修课教材,也可供有关工程技术人员从事工程设计、地基基础计算时参考。

地 基 基 础 计 算 与 程 序 设 计

王国体 主编

*

中国科学技术大学出版社出版

(安徽省合肥市金寨路96号 邮政编码: 230026)

合肥市东方印刷厂印刷

安徽省新华书店发行

*

开本: 787×1092/16 印张: 15 字数: 350千

1991年2月第1版 1991年2月第1次印刷

印数: 1—3000册

ISBN7-312-00283-8/TU·1 定价: 6.50元

前 言

八十年代以来,随着计算机特别是微型机的应用和发展,地基基础这一古老而又新兴的学科焕发出蓬勃生机。土力学方面的数值计算、基础设计 CAD 系统的研制和土工测试的自动控制与数据处理,构成了计算机在该学科应用的三个基本领域。在数值计算领域中,六十年代以来发展起来的工程数值计算方法,如有限元法、边界元法等,只是依赖于计算机才得以真正地应用于工程实践。也只有利用计算机,才使得基础设计的优化和土工测试的自动控制成为现实。

这几年来,国内曾先后召开了涉及本学科的全国性、区域性计算机应用学术交流会,对计算机在本学科的科学研究的、工程技术、高等教育领域的应用起了一定的推动作用。目前,一些高等工科院校在本学科的教学,为个别专业的高年级学生开设了计算机应用及程序设计课程,但国内还没有出版过这方面的著作和教学参考书。为了工程设计及教学的需要,我们特编著本书,和读者一起,为共同提高计算机在本学科的应用水平而努力。

本书在介绍计算理论方法上力求简明扼要,着重突出应用程序设计,使那些学过计算机程序语言的人经过简单的学习就可以上机操作;考虑到有限元法对一般工程技术人员来说仍然是比较新的数值分析方法,对这部分理论计算公式作了简要的推导。全书分九章:第一章,地基土室内常规试验数据整理与作图应用程序,介绍了颗粒分析试验、液塑限联合测定试验、侧限压缩试验及抗剪强度试验等数据处理及作图七个应用程序;第二章,地基应力与基础的沉降计算及程序,包括分层总和法和《建筑地基基础设计规范》方法及考虑相邻基础影响时求基础的最终沉降量的三个计算程序;第三章,地基承载力的计算与程序,提供了地基承载力的各家理论计算公式及相应的计算程序;第四章,粘性土坡稳定分析计算与程序,以条分法的太沙基公式、简化毕肖甫公式为基础,提供了包括考虑稳定渗流作用下土坡或坝的稳定分析程序;第五章,按沉降要求基础群底面积的计算与程序,对群基础设计中按等沉降(包括沉降差)和控制某一沉降为条件,计算多个基础底面积的应用程序;第六章,重力式挡土墙计算与程序,以土压力的库伦理论、朗金理论为基础,对工程中常见的各种挡土型式的挡土墙的土压力数值和抗滑、抗倾覆稳定进行计算的应用程序;第七章,弹性地基梁有限元法计算与程序,对有限元法计算基本公式作了简要的推导,对单元刚度矩阵、总刚度矩阵的形成作了实用上的说明,提供了等厚度、变厚度的文克尔地基梁和格式条形基础的计算程序;第八章,地基中墙柱有限元法计算与程序,对地基内竖向构件有限元计算的基本公式、单元刚度矩阵、总刚度矩阵作了必要的推导与说明,提供了抗滑坡刚架有限元法计算程序;第九章,弹性地基梁和板的有限差分法计算与程序,对文克尔地基上梁及板有限差分法计算公式作了简要介绍,提供了梁、板差分计算程序。附录一介绍了本书使用的矩阵符号和三角化法及有限元法计算时的存储技巧,附录二列出了本书使用的 BASIC 语言的语句和函数。全书提供的应用程序通用实用,并附有算例,对于实际计算问题只需按所给格式或参照算例填入计

算数据，经上机即可得到满意的计算结果。

本书可作为高等工科院校的土木、建筑、交通、水利等专业师生的教学参考书、上机实践指导书或选修课教材，也可供有关工程技术人员从事工程设计、地基基础计算时参考。

本书由合肥工业大学王国体主编，由肖运龙（武汉城市建设学院）、王志建（安徽建筑工业学院）、王正秋（哈尔滨建筑工程学院）等任副主编，编委有：金康宁（武汉城市建设学院）、周书兰（黄河大学）、吴雄志（河北煤炭建筑工程学院）、方达宪（合肥工业大学）、肖仁成（衡阳工学院）、王 园（新疆八一农学院）、刘文白（内蒙古工学院）等。

由于作者水平所限，书中难免存在缺点和不妥之处，诚恳希望读者批评指正。

王国体

1990年10月

目 录

前 言	(I)
第一章 地基土室内常规试验数据整理与作图应用程序	(1)
第一节 颗粒分析试验数据整理及作图应用程序	(1)
第二节 液塑限联合测定试验计算与作图应用程序	(7)
第三节 侧限压缩试验计算与作图应用程序	(10)
第四节 抗剪强度试验计算与作图应用程序	(15)
第五节 抗剪强度参数规范方法计算与作图应用程序	(26)
第二章 地基应力与基础的沉降计算及程序	(33)
第一节 地基竖向应力计算公式	(33)
第二节 分层总和法求基础的最终沉降量	(35)
第三节 基础沉降量的分层总和法计算程序	(36)
第四节 相邻基础影响时求最终沉降量计算程序	(42)
第五节 《建筑地基基础设计规范》法最终沉降量计算程序	(49)
第三章 地基承载力的计算与程序	(60)
第一节 地基承载力计算的理论公式	(60)
第二节 程序与算例	(63)
第四章 粘性土坡稳定分析计算与程序	(67)
第一节 粘性土坡稳定分析方法	(67)
第二节 程序设计要点	(68)
第三节 土坡稳定分析程序	(70)
第四节 稳定渗流对土坡(坝)影响计算与程序	(77)
第五章 按沉降要求基础群底面积的计算与程序	(86)
第一节 基本概念与公式	(86)
第二节 程序设计与框图	(89)
第三节 应用算例和源程序	(90)
第六章 重力式挡土墙计算与程序	(102)
第一节 土压力计算理论	(102)
第二节 工程中的重力式挡土墙土压力计算	(104)
第三节 重力式挡土墙设计程序	(109)
第四节 算例及源程序	(111)
第七章 弹性地基梁有限元法计算与程序	(121)
第一节 弹性地基构件概述	(121)
第二节 弹性地基梁挠曲线方程及其解答	(122)

第三节	弹性地基梁的扭转·····	(126)
第四节	弹性地基梁的有限元分析·····	(128)
第五节	条形基础计算程序·····	(135)
第六节	格式条形基础的计算与程序·····	(143)
第八章	地基内墙柱有限元法计算与程序·····	(159)
第一节	地基系数 k 的计算选取·····	(159)
第二节	地基内竖向基础构件的挠曲线方程及其解答·····	(160)
第三节	地基内墙柱单元分析·····	(163)
第四节	抗滑坡刚架的整体分析·····	(167)
第五节	抗滑坡刚架有限元法计算程序·····	(175)
第九章	弹性地基梁和板的有限差分法计算与程序·····	(189)
第一节	有限差分法一般原理和公式·····	(189)
第二节	弹性地基梁的有限差分法计算与程序·····	(191)
第三节	弹性地基矩形板的有限差分法计算与程序·····	(205)
附 录		
一、	矩阵符号及有关运算·····	(225)
二、	BASIC 函数与语句·····	(229)
参考文献	·····	(230)

第一章 地基土室内常规试验数据整理 与作图应用程序

在建筑物的地基勘察中，通常采用现场原位测试手段与室内土工试验相结合，提供地基的物理、力学性质参数，对地基进行分析、评价，为建筑物的基础选址、平面布置、地基与基础设计及合理施工提供必要的资料。因此，室内试验数据整理的有关参数，对建筑工程的设计、施工、使用有着其重要的意义。为了克服在数据整理中的人为因素，采用计算机处理是十分有效的。

本章据室内部份常规试验，对较复杂试验数据的整理提供计算机应用程序，它们是：地基土颗粒分析试验、液、塑限联合测定试验、侧限压缩试验和抗剪强度试验数据整理和作图应用程序。

第一节 颗粒分析试验数据整理及作图应用程序

一、颗粒分析试验的数据准备

地基土是由大小不同的土粒组成的。土的粒径级配是反映天然土体中各粒组相对含量的不同所构成的基本特性。

土体的粒组划分见表1-1，表1-2。工程上常用的粒径分析方法是筛分法(适用粗粒土)和水分法—比重计法和吸管法(适用于细粒土)，由于地基土的粒径分布较广，通常要联合采

表1-1 建筑地基基础设计规范 (GBJ7-89)

粒组 名称	漂石	卵石	圆砾	砂 粒				粉粒	粘 粒
	碎石	碎石	角砾	粗	中	细	粉		
粒径 范围 mm	>200	200~20	20~2	2~0.5	0.5~0.25	0.25~0.075	0.075~0.01	0.01~0.005	<0.005

表1-2 土工试验规程 (SDS01-79) 土-001-79

粒组 名称	粗 粒 组									细 粒 组	
	漂石	碎石	砾			砂 粒				粉 粒	粘 粒
	块石	卵石	粗	中	细	粗	中	组	极细		
粒径 范围 mm	>300	200 } 60	60 } 20	20 } 5	5 } 2	2 } 0.5	0.5 } 0.25	0.25 } 0.1	0.1 } 0.05	0.05~0.005	<0.005

用。将在上述试验中得到的数据换算为小于某粒径的颗粒占整个试验土样重的百分数，如表1-3，在以对数尺度表示粒径(d_{mm})为横坐标，算术尺度表示小于某粒径的含量占总土重的百分数 P 为纵坐标的坐标图中，将表1-3中试验点描绘成粒径级配曲线（如图1-2）。据粒径曲线的缓陡可以定性的判定土样颗粒级配的优劣，而定量表示可采用不均匀系数 C_u 和曲率系数 C_c 的值来判定。

表1-3 粒径分析试验数据

粒径 (mm)	20	10	5	3	1	0.5	0.25	0.1	0.05	0.01	0.005
小于该粒径 的颗粒占土 重的%	100	99	96	91	84	64	36	17	11	5	2

$$C_u = \frac{d_{60}}{d_{10}} \quad (1-1)$$

$$C_c = (d_{30})^2 / (d_{10} \cdot d_{60}) \quad (1-2)$$

式中： d_{60} 、 d_{10} 、 d_{30} 分别表示级配曲线上小于某粒径的含量分别是60%、10%、30%时所对应的粒径(mm)， d_{10} 称为有效粒径， d_{60} 称为控制粒径。

不均匀系数 C_u 和曲率系数 C_c 的大小反映了土体中土粒大小分布的离散性、连续性及均匀性，工程上一般认为 $C_u \geq 5$ 且 $C_c = 1 \sim 3$ 时，土体的颗粒级配良好，否则土的级配不良。

二、粒径级配曲线绘制与计算程序设计

1. 半对数坐标图的绘制

一般计算机的屏幕使用分文本状态和图形状态。文本状态下以显示字符为基本元素，每帧 $\times \times$ 行，每行 $\times \times$ 字符〔注〕。图形状态下，屏幕显示以点为基本单元，点的多少由分辨率来确定（见附录2）。文本状态和图形状态之间的转换可以由有关指令来实现。

计算机的屏幕可以看成由 $x \sim y$ 平面直角坐标系构成的图面，不过它的原点在左上角， y 方向以向下为正，有些机器系统可以通过使用有关指令改变原点和坐标系统。通过选择不同屏幕背景和色笔来实现作图。

设图形状态下， x 方向的坐标点数为 F_x ， y 方向为 F_y ，则整个作图图幅为 $F_x \times F_y$ 个点，对于作颗粒级配曲线，首先应根据粒径范围把它限定在 F_x 内， F_y 为累积百分数。若粒径范围为 $x_0 \text{mm} \sim x_9 \text{mm}$ ，($x_0 > x_9$)，则 F_x 应划分为 N 段，每段的长度为 W_x ，则，

$$\left. \begin{aligned} N &= F_x / [\log(x_0) - \log(x_9)] \\ W_x &= F_x / N \end{aligned} \right\} \quad (1-3)$$

每段长度在对数值上为单位1，这样经过程序处理可以绘成粒径在 $x_0 \sim x_9$ 的范围内的半对数坐标网格图。

2. 级配曲线的描绘

当粒径与其相应占总重百分数的数据关系确定之后，在半对数坐标系中就可以确定对应的点，使用计算机采用数值插值将它们描绘成光滑曲线，通常采用抛物插值或三次样条函数

注：IBMPC、Apple、长城为25行，每行80字符、（西文方式），PC-1500为1行19个字符。

插值拟合，下面给出抛物线插值公式：

$$\begin{aligned}
 x_d(P) = & \frac{(P-P_i)(P-P_{i+1})}{(P_{i-1}-P_i)(P_{i-1}-P_{i+1})} \cdot x_{di-1} \\
 & + \frac{(P-P_{i-1})(P-P_{i+1})}{(P_i-P_{i-1})(P_i-P_{i+1})} \cdot x_{di} \\
 & + \frac{(P-P_{i-1})(P-P_i)}{(P_{i+1}-P_{i-1})(P_{i+1}-P_i)} \cdot x_{di+1} \quad (1-4)
 \end{aligned}$$

$$P_{i-1} < P \leq P_i$$

x_d —某插值粒径，mm；

P —小于 x_d 粒径的重量占总重的百分数；

x_{di} , P_i ; x_{di-1} , P_{i-1} ; x_{di+1} , P_{i+1} 的关系同 x_d 与 P 。

这样，每一组对应数据在屏幕上的坐标位置是：

$$\begin{cases} x = F_x - W_x \cdot \log(x_d/x_9) \\ y = F_y \cdot P/Y_9 \end{cases} \quad (1-5)$$

3. 程序设计

(1) 变量说明

输入变量：

xz , xu ——屏幕作图左、右空余点列数（尺寸）；

yz , yu ——屏幕作图上、下空余点行数（尺寸）；

F_x , F_y ——屏幕作图 x 方向， y 方向点数，受作图分辨率控制，见附录二；

x_0 , x_9 ——粒径分布范围的左、右控制尺寸，mm，（取对数后应为整数）；

y_0 , y_9 ——百分数范围（0, 100）；

N_x —— x 对数坐标单位长度数，见式（1-3）；

N_y —— y 坐标分段数（以10%划分， $N_y=10$ ，；以20%划分， $N_y=5$ ）

ND ——试验数据对应点数；

ys ——绘曲线插值步长（取1）；

$XN(ND)$, $YN(ND)$ ——粒径数组与其对应的百分数数组。

输出变量

$D-60$, $D-30$, $D-10$ （程序过程中为 D_6 , D_3 , D_1 ）同前 d_{60} , d_{30} , d_{10} ；

CU , CC 同前 C_u , C_o 。

(2) 数据输入采用数据表形式（即使用DATA语句）

数据表形式如下：

〈行号〉DATA 〈 xz 〉, 〈 xu 〉, 〈 yz 〉, 〈 yu 〉, 〈 F_x 〉, 〈 F_y 〉

〈行号〉DATA 〈 x_0 〉, 〈 x_9 〉, 〈 y_0 〉, 〈 y_9 〉, 〈 N_x 〉, 〈 N_y 〉

〈行号〉DATA 〈 ND 〉, 〈 d_1 〉, 〈 P_1 〉, ……, 〈 d_{ND} 〉, 〈 P_{ND} 〉

〈行号〉DATA 〈 ys 〉

(3) 程序结构框图见图1-1。

三、算例

采用表1-3中的试验数据, 填写数据表如下:

10 DATA 5,5,5,5,5,319,199

20 DATA 100,0.001,0,100,5,10

30 DATA 11,20,100,10,99,5,96,2,91,1,
84,0.5,64,0.25,36,0.1,17,0.05,
11,0.01,5,0.005,2

40 DATA 1

(Nx, 在填写时, 对数以10为底)

运行结果如下:

* D-60=.39

* D-30=.08

* D-10=.01

* CU--=39

* CC--=1.64

颗粒累积曲线图见图1-2。




图1-1




图1-2 计算机屏幕输出或打印绘出曲线图

四、源程序 (program-1)

```

100 REM *****
105 REM * PROGRAM OF GRAIN SIZE ACCUMULATION CURVE *
110 REM *****
115 REM *****
120 REM * MAIN PROGRAM *
125 REM *****
  
```

```

130 REM.....DATA PREPARED.....
135 READ XZ,XU,YZ,YU,FX,FY,X0,X9,Y0,Y9,NX,NY
140 READ ND
145 FX=FX-XZ-XU
150 FY=FY-YZ-YU
155 WX=FX/NX:WY=FY/NY
160 XX=SGN(LOG(X9)/LOG(10)-LOG(X0)/LOG(10))
165 YY=(Y9-Y0)/NY
170 DIM XN(ND),YN(ND)
175 FOR I=1 TO ND
180 READ XN(I),YN(I)
185 PRINT XN(I),YN(I)
190 NEXT I
195 IF XN(1)>X0 OR XN(ND)<X9 THEN PRINT" X...DATA ERROR"
200 IF YN(1)>Y9 OR YN(ND)<Y0 THEN PRINT" Y...DATA ERROR"
205 REM.....DRAW COORDINATE.....
210 CLS
215 SCREEN 1:COLOR 7,0
220 FOR I=NX TO 1 STEP -1
225 XP=X0*10^(XX*(I-1))
230 X1=I*WX+XZ
235 LINE (X1,FY+YZ)-(X1+3,FY+YZ),3
240 FOR J=1 TO 10
245 X=X1-WX*LOG(J)/LOG(10)
250 LINE (X,FY+YZ)-(X,YZ),1
255 NEXT J
260 NEXT I
265 FOR I=0 TO NY
270 Y1=YZ+I*WY
275 YP=Y9+YY*I
280 LINE (0,Y1)-(XZ,Y1),3
285 LINT (XZ,Y1)-(FX+XZ,Y1),1
290 NEXT I
295 REM .....DRAM ACCUMULATION CURVE.....
300 FOR I=1 TO ND
305 XN(I)=FX-WX*LOG(XN(I)/X9)/LOG(10)+XZ
310 YN(I)=FY-FY*YN(I)/Y9+YZ
315 CIRCLE (XN(I),YN(I)),1,3,,,1

```

```

320 NEXT I
325 X1=XN(1):Y1=YN(1):LINE (X1,Y1)-(X1,Y1)
330 READ YS
335 Y=Y1+YS
340 IF Y>YN(ND) THEN 395
345 FOR J=1 TO ND
350 IF ABS (Y-YN(J))<YS THEN X=XN(J):Y=YN(J):GOTO 370
355 NEXT J
360 GOSUB 500
365 X=EE
370 IF ABS(Y-FY*.4-YZ)<=YS THEN D6=X-XZ
375 IF ABS(Y-FY*.7-YZ)<=YS THEN D3=X-XZ
380 IF ABS(Y-FY*.9-YZ)<=YS THEN D1=X-XZ
385 LINE -(X,Y),4
390 X1=X:Y1=Y:GOTO 335
395 LINE (XZ,YZ+FY*.4)-(D6+XZ,YZ+FY*.4),7:LINE -(D6+
    XZ,YZ+FY),5
400 LINE (XZ,YZ+FY*.7)-(D3+XZ,YZ+FY*.7),5:LINE -(D3+
    XZ,YZ+FY),5
405 LINE (XZ,YZ+FY*.9)-(D1+XZ,YZ+FY*.9),5:LINE -(D1+
    XZ,YZ+FY),5
410 STOP
415 CLS:SCREEN 0
420 LPRINT "....."
425 D6=(FX-D6)/WX:DD6=10^D6*X9:DD6=INT (DD6*100+.5)/100
430 LPRINT " * D-60=",DD6
435 D3=(FX-D3)/WX:DD3=10^D3*X9:DD3=INT(DD3*100+.5)/100
440 LPRINT " * D-30=",DD3
445 D1=(FX-D1)/WX:DD1=10^D1*X9:DD1=INT(DD1*100+.5)/100
450 LPRINT " * D-10=",DD1
455 LPRINT "....."
460 LPRINT :LPRINT
465 IF DD1=0 OR DD3=0 THEN 495
470 CU=INT(DD6/DD1*100+.5)/100
475 CC=INT(DD3*DD3/DD6/DD1*100+.5)/100
480 LPRINT " * CU--=",CU
485 LPRINT " * CC--=",CC
490 LPRINT "....."

```

```

495 END
500 REM *****
505 REM *          SUBPROGRAM -- INSET VALAUE          *
510 REM *****
515 IF Y<YZ(1) THEN CC=1 :GOTO 540
520 IF Y>YN(ND) THEN CC=ND-2:GOTO 540
525 FOR K=1 TO ND-2
530 IF Y>=YN(K) AND Y<=YN(K+1) THEN CC=K:GOTO 540
535 NEXT K
540 X1=XN(CC):Y1=(CC)
550 X2=XN(CC+1):Y2=YN(CC+1)
555 X3=XN(CC+2):Y3=YN(CC+2)
560 EE=(Y-Y1)*(Y-Y2)/(Y3-Y2)/(Y3-Y1)*X3
565 EE=(Y-Y1)*(Y-Y3)/(Y2-Y3)/(Y2-Y1)*X2+EE
570 EE=(Y-Y2)*(Y-Y3)/(Y1-Y2)/(Y1-Y3)*X1+EE
575 RETURN

```

第二节 液塑限联合测定试验计算与作图应用程序

一、液限、塑限的联合测定

液限、塑限的联合测定试验是用圆锥仪测定土在不同含水量时圆锥在规定时间（15秒钟）的入土深度在双对数坐标中入土深度与含水量成直线关系而设计的。入土深度为10mm时相应含水量称为液限，入土深度为2mm时相应含水量称为塑限。一般试验制备三种含水量的土膏，通过作图求出液限、塑限。


三种含水量与相应的入土深度数据在双对数坐标系中应接近一直线，若不在一直线上，通过高含水量的一点与其余两点连两条直线，在圆锥入土深度为2mm处求得两个相应的含水量，这两个含水量的差值不应超过2%。否则，另外补土样做试验。

二、程序设计

1. 程序结构整体框图（图1-3）

2. 变量说明：

L1, W1; L2, W2; L3, W3——圆锥仪不同的入土深度与其对应的含水量，L为mm，



W去掉%;

xz,xu,yz,yu,Fx,Fy同前。

WP, WL——误差在2%以内求得的塑限和液限;

WP, WZ——误差>2%时塑限变化的可能范围值;

WL, WM——误差>2%时液限变化的可能范围值。

3.数据表形式

<行号> DATA <L1>,<W1>,<L2>,<W2>,<L3>,<W3>

<行号> DATA <xz>,<xu>,<yz>,<yu>,<Fx>,<Fy>

注意L1,L2,L3输入时应满足: $L1 < L2 < L3$, 否则重新填写数据。

三、算例:

某地基三个土样的液塑限联合测定结果如下(表1-4)。

表1-4

土 样	I /			II			III		
入土深度	L1	L2	L3	L1	L2	L3	L1	L2	L3
mm	9.8	7.4	3.4	9.1	5.6	3.5	9.8	7.8	3.2
含水量 %	33.4	31.1	25.7	26.0	23.4	21.2	44.8	40.8	30.7

数据段如下:(图形输出从略)

I:

10 DATA 3.4,25.7,7.4,31.1,9.8,33.4

20 DATA 10,10,10,10,319,199

输出

* PLASTIC LIMIT == 22.42

* LIQUIT LIMIT == 33.57

II:

10 DATA 3.5,21.2,5.6,23.4,9.1,26.0

20 DATA 25,25,25,25,190,190

输出

* PLASTIC LIMIT == 18.76

* LIQUIT LIMIT == 26.53

若 $L3=9.1, W3=26.4$

10 DATA 3.5,20.2,5.6,23.4,9.1,26.4

20 DATA 125,00,25,25,25,319,190

则输出:

PLASTIC LIMIT = 17.42 -- 18.71

LIQUIT LIMIT = 26.66 -- 24.94

TEST AGAIN PLEASE

四、源程序 (program-2)

```
110 REM * * * * *
115 READ L1,W1,L2,W2,L3,W3
120 READ XZ,XU,YZ,YU,FX,FY
125 IF L1<L2 AND L2<L3 AND W1<W2 AND W2<W3 THEN 142
130 PRINT"CHECKE INPUT DATA!":PRINT:PRINT
135 PRINT"MUST BE L1<L2<L3 AND W1<W2<W3!"
140 PRINT:PRINT:GOTO 390
142 CLS
145 SCREEN 1:COLOR 7,0
150 FX=FX-XZ-XU:FY=FY-YZ-YU
155 FOR I=1 TO 10
160 Y=FY-FY*LOG(I)/LOG(10)+YZ
165 LINE(XZ,Y)-(FX+XZ,Y),1
170 NEXT I
175 FOR I=1 TO 10
180 X=XZ+FX*LOG(I)/LOG(10)
185 LINE(X,YZ)-(X,YZ+FY),1
190 NEXT I
195 Y1=LOG(L1)/LOG(10)*FY
200 Y2=LOG(L2)/LOG(10)*FY
205 Y3=LOG(L3)/LOG(10)*FY
210 X1=LOG(W1/10)/LOG(10)*FX
215 X2=LOG(W2/10)/LOG(10)*FX
220 X3=LOG(W3/10)/LOG(10)*FX
225 CIRCLE(X1+XZ,FY-Y1+YZ),1,3,,,1
230 CIRCLE(X2+XZ,FY-Y2+YZ),1,3,,,1
235 CIRCLE(X3+XZ,FY-Y3+YZ),1,3,,,1
240 Y=LOG(2)/LOG(10)*FY
245 XP=(X3-X1)/(Y3-Y1)*(Y-Y1)+X1
250 XY=(X3-X2)/(Y3-Y2)*(Y-Y2)+X2
255 LINE(X1+XZ,FY-Y1+YZ)-(X3+XZ,FY-Y3+YZ),3
260 LINE(X2+XZ,FY-Y2+YZ)-(X3+XZ,FY-Y3+YZ),3
265 WZ=EXP(XP/FX*LOG(10))*10
270 WP=EXP(XY/FX*LOG(10))*10
```

```

275 IF ABS(WP-WZ)/WP<=.02 THEN 335
280 XL=(X3-X1)/(Y3-Y1)*(FY-Y1)+X1
285 XM=(X3-X2)/(Y3-Y1)*(FY-Y2)+X2
290 WL=EXP(XL/FX*LOG(10))*10
295 WY=EXP(XM/FX*LOG(10))*10
300 WZ=INT(WZ*100+.5)/100
305 WP=INT(WP*100+.5)/100
310 WL=INT(WL*100+.5)/100
315 WY=INT(WY*100+.5)/100
320 LPRINT"PLASTIC LIMIT=",WZ,"--",WP
325 LPRINT"LIQUIT LIMIT=",WL,"--",WY
330 LPRINT"TEST AGAIN PLEASE!"
335 WP=(WP+WZ)/2
340 XL=(X3-X1)/(Y3-Y1)*(FY-Y1)+X1
345 X=LOG(WP/10)/LOG(10)*FX
350 LINE(X+XZ,FY-Y+YZ)-(XL+XZ,YZ),1
355 WL=EXP(XL/FX*LOG(10))*10
360 WP=INT(WP*100+.5)/100
365 WL=INT(WL*100+.5)/100
370 LPRINT"-----"
375 LPRINT"* PLASTIC LIMIT ==",WP
380 LPRINT"* LIQUIT LIMIT ==",WL
385 LPRINT"-----"
390 END

```

第三节 侧限压缩试验计算与作图应用程序

一、压缩试验的数据准备

在室内侧限压缩试验时，将观测数据进行整理，可以得到各级压力 P_i 作用下土样的最终压缩量 S_i 并求出这时土样的孔隙比 e_i ，

$$e_i = e_0 - \frac{S_i}{H_0} (1 + e_0) \quad (1-6)$$

式中， e_0 ——初始孔隙比，可以由室内其他试验求出， $e_0 = \frac{r_w G_s (1+w)}{r} - 1$ ， r_w, G_s ，

w, r 分别为水的重度(kN/m^3)，土粒比重，天然含水量(%)，和土天然重度(kN/m^3)。

H_0 ——土样的初始高度，mm；一般 $H=20\text{mm}$ ，各级压力间的压缩系数 $a_{(1-1,1)}$ 为，

$$a_{(1-1,1)} = \frac{e_{i-1} - e_i}{p_i - p_{i-1}} \quad (1-7)$$